Heterogeneous Variance Models from Hoffman Chapter 8: Predicting Differential Variation in Fluctuation of Daily Physical Symptoms from Sex, Age, and Daily Negative Mood (PMC) and Stressor Days (GMC)

Same baseline homogeneous variance model via SAS PROC MIXED or via SAS PROC NLMIXED:

```
TITLE1 "Baseline Homogeneous Variance Model with just Fixed Effects";
PROC MIXED DATA=example COVTEST NOCLPRINT NAMELEN=100 IC METHOD=ML;
CLASS ID;
MODEL symptom5 = female age80 female*age80 PMmood2 PMstress40 female*PMstress40 PMmood2*PMmood2
                 WPmood PMmood2*WPmood TVstress0 female*TVstress0 / SOLUTION DDFM=SATTERTHWAITE;
RANDOM INTERCEPT / TYPE=UN SUBJECT=ID;
REPEATED / TYPE=VC SUBJECT=ID; RUN; TITLE1;

TITLE1 "Baseline Homogeneous Variance Model with just Fixed Effects";
PROC NLMIXED DATA=example METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS  fint=2.45 fsex=-.33 fage=.07 fsexage=-.09
       fbpmood=-.34 fwpmood=.26 fcstress=1.93 fwpstress=.22
       f2sexstress=-1.33 f1sexstress=-.18 fbpmood2=-1.78 fbpwpmood=.26
       vU0int=-.47 varEint=-.50;
* Setting up level-2 equations;
       b0i = fint + fsex*female + fage*age80 + fsexage*female*age80
             + fbpmood*PMmood2 + fcstress*PMstress40
             + f2sexstress*PMstress40*female + fbpmood2*PMmood2*PMmood2 + U0i;
       b1i = fwpmood + fbpwpmood*PMmood2;
       b2i = fwpstress + f1sexstress*female;
* Setting up level-1 equation WITHOUT level-1 residual;
       y = (b0i) + (b1i*WPmood) + (b2i*TVstress0);
* Model for heterogeneous variance at level-2 and level-1;
       vU0 = EXP(vU0int);
       varE = EXP(varEint);
* Telling it which DV, defining level-1 residual;
* symptom5 is normally distributed with a mean of "y" and a variance of "VarE";
       MODEL symptom5 ~ normal(y,varE);
* Defining random effects: normally distributed with means and variances;
       RANDOM U0i ~ normal([0],[vU0]) SUBJECT=ID;
* Requesting additional model-implied estimates;
       ESTIMATE "Actual Random Intercept Variance" EXP(vU0int);
       ESTIMATE "Actual Residual Variance"         EXP(varEint);
RUN; TITLE1;
```

Level 1 :

$$\text{Symptoms}_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Mood}_{ti} - \overline{\text{Mood}}_i\right) + \beta_{2i}\left(\text{Stressor}_{ti}\right) + e_{ti}$$

Residual Variance: $\sigma^2_{e_{ti}} = \exp\left[\eta_{0i}\right]$

Level 2 :

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}\left(\text{Women}_i\right) + \gamma_{02}\left(\text{Age}_i - 80\right) + \gamma_{03}\left(\text{Women}_i\right)\left(\text{Age}_i - 80\right) + \gamma_{04}\left(\overline{\text{Mood}}_i - 2\right)$$
$$+ \gamma_{08}\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{09}\left(\text{Women}_i\right)\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{0,16}\left(\overline{\text{Mood}}_i - 2\right)^2 + U_{0i}$$

Within-Person Mood: $\beta_{1i} = \gamma_{10} + \gamma_{14}\left(\overline{\text{Mood}}_i - 2\right)$

Within-Person Stressor: $\beta_{2i} = \gamma_{20} + \gamma_{21}\left(\text{Women}_i\right)$

Random Intercept Variance $\tau^2_{U_{0i}} = \exp\left[\upsilon_{00}\right]$

Residual Variance: $\eta_{0i} = \varepsilon_{00}$

PROC MIXED OUTPUT:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	id	0.6230	0.1044	5.96	<.0001	Random Intercept Variance
Residual	id	0.6093	0.04292	14.19	<.0001	Residual Variance

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
1379.2	**14**	1407.2	1408.1	1422.3	1444.4	1458.4

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	2.4473	0.3474	117	7.05	<.0001
female	-0.3277	0.2140	146	-1.53	0.1278
age80	0.07311	0.03019	109	2.42	0.0171
female*age80	-0.09127	0.03381	108	-2.70	0.0081
PMmood2	-0.3396	0.7999	110	-0.42	0.6720
PMstress40	1.9322	0.6005	124	3.22	0.0016
female*PMstress40	-1.3281	0.6617	128	-2.01	0.0468
PMmood2*PMmood2	-1.7784	0.7131	108	-2.49	0.0141
WPmood	0.2634	0.2213	405	1.19	0.2346
PMmood2*WPmood	0.2591	0.3962	405	0.65	0.5135
TVstress0	0.2192	0.1710	404	1.28	0.2007
female*TVstress0	-0.1812	0.2053	404	-0.88	0.3780

PROC NLMIXED OUTPUT:

Dimensions

Observations Used	509
Observations Not Used	0
Total Observations	509
Subjects	105
Max Obs Per Subject	5
Parameters	**14**
Quadrature Points	1

Specifications

Data Set	WORK.EXAMPLE
Dependent Variable	symptom5
Distribution for Dependent Variable	Normal
Random Effects	UOi
Distribution for Random Effects	Normal
Subject Variable	id
Optimization Technique	Newton-Raphson
Integration Method	Adaptive Gaussian Quadrature

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	**1379.2**
AIC (smaller is better)	1407.2
AICC (smaller is better)	1408.1
BIC (smaller is better)	1444.4

Make sure the gradient (partial first derivative with respect to each parameter) is as close to 0 as possible. Otherwise, the estimation really hasn't finished on the "best" estimates.

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper	**Gradient**
fint	2.4473	0.3474	104	7.05	<.0001	0.05	1.7585	3.1362	-6.09E-8
fsex	-0.3277	0.2140	104	-1.53	0.1286	0.05	-0.7520	0.09659	3.344E-8
fage	0.07311	0.03019	104	2.42	0.0172	0.05	0.01324	0.1330	5.738E-6
fsexage	-0.09127	0.03381	104	-2.70	0.0081	0.05	-0.1583	-0.02423	3.875E-6
fbpmood	-0.3396	0.7999	104	-0.42	0.6720	0.05	-1.9257	1.2466	1.156E-7
fwpmood	0.2634	0.2213	104	1.19	0.2366	0.05	-0.1754	0.7022	-1.09E-8
fcstress	1.9322	0.6005	104	3.22	0.0017	0.05	0.7415	3.1230	1.034E-8
fwpstress	0.2192	0.1710	104	1.28	0.2028	0.05	-0.1199	0.5584	-6.11E-9
f2sexstress	-1.3281	0.6617	104	-2.01	0.0473	0.05	-2.6402	-0.01600	1.161E-8
f1sexstress	-0.1812	0.2053	104	-0.88	0.3796	0.05	-0.5882	0.2259	3.322E-8
fbpmood2	-1.7784	0.7131	104	-2.49	0.0142	0.05	-3.1926	-0.3643	-9.03E-8
fbpwpmood	0.2591	0.3962	104	0.65	0.5145	0.05	-0.5266	1.0449	7.878E-9
vUOint	-0.4733	0.1677	104	-2.82	0.0057	0.05	-0.8057	-0.1408	-5.58E-7
varEint	-0.4955	0.07045	104	-7.03	<.0001	0.05	-0.6352	-0.3558	-7.07E-7

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Actual Random Intercept Variance	**0.6230**	**0.1044**	**104**	**5.96**	**<.0001**	**0.05**	**0.4159**	**0.8301**
Actual Residual Variance	**0.6093**	**0.04292**	**104**	**14.19**	**<.0001**	**0.05**	**0.5241**	**0.6944**

Testing a ω_{0i} scale factor for random individual differences in within-person variation:

Level 1:

$$\text{Symptoms}_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Mood}_{ti} - \overline{\text{Mood}}_i\right) + \beta_{2i}\left(\text{Stressor}_{ti}\right) + e_{ti}$$

$$\text{Residual Variance: } \sigma^2_{e_{ti}} = \exp\left[\eta_{0i}\right]$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}\left(\text{Women}_i\right) + \gamma_{02}\left(\text{Age}_i - 80\right) + \gamma_{03}\left(\text{Women}_i\right)\left(\text{Age}_i - 80\right) + \gamma_{04}\left(\overline{\text{Mood}}_i - 2\right)$$
$$+ \gamma_{08}\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{09}\left(\text{Women}_i\right)\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{0,16}\left(\overline{\text{Mood}}_i - 2\right)^2 + U_{0i}$$

$$\text{Within-Person Mood: } \beta_{1i} = \gamma_{10} + \gamma_{14}\left(\overline{\text{Mood}}_i - 2\right)$$

$$\text{Within-Person Stressor: } \beta_{2i} = \gamma_{20} + \gamma_{21}\left(\text{Women}_i\right)$$

$$\text{Random Intercept Variance } \tau^2_{U_{0i}} = \exp\left[\upsilon_{00}\right]$$

$$\text{Residual Variance: } \eta_{0i} = \varepsilon_{00} + \omega_{0i}$$

```
TITLE1 "Baseline Model with just Fixed Effects";
TITLE2 "Test Scale Factor";
PROC NLMIXED DATA=example METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;  *METHOD=FIRO;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS  fint=2.45 fsex=-.33 fage=.07 fsexage=-.09
       fbpmood=-.34 fwpmood=.26 fcstress=1.93 fwpstress=.22
       f2sexstress=-1.33 f1sexstress=-.18 fbpmood2=-1.78 fbpwpmood=.26
       varEint=-.50 vU0int=-.47 vscale=-.22 cU0scale=.00;
* Setting up level-2 equations;
       b0i = fint + fsex*female + fage*age80 + fsexage*female*age80
             + fbpmood*PMmood2 + fcstress*PMstress40
             + f2sexstress*PMstress40*female + fbpmood2*PMmood2*PMmood2 + U0i;
       b1i = fwpmood + fbpwpmood*PMmood2;
       b2i = fwpstress + f1sexstress*female;
* Setting up level-1 equation WITHOUT level-1 residual;
       y = (b0i) + (b1i*WPmood) + (b2i*TVstress0);
* Model for heterogeneous variance at level-2 and level-1;
       vU0 = EXP(vU0int);
       varE = EXP(varEint + Scalei);
* Telling it which DV, defining level-1 residual;
* symptom5 is normally distributed with a mean of "y" and a variance of "VarE";
       MODEL symptom5 ~ normal(y,varE);
* Defining random effects: normally distributed with means and variances;
       RANDOM U0i Scalei ~ normal([0,0],[vU0,cU0scale,vscale]) SUBJECT=ID;
RUN; TITLE1; TITLE2;
```

PROC NLMIXED OUTPUT:

```
                        Specifications
Data Set                                    WORK.EXAMPLE
Dependent Variable                          symptom5
Distribution for Dependent Variable         Normal
Random Effects                              U0i Scalei
Distribution for Random Effects             Normal
Subject Variable                            id
Optimization Technique                      Newton-Raphson
Integration Method                          Adaptive Gaussian Quadrature
```

That's all we get – SAS could not find an estimate for the scale factor variance, even after switching to less intensive estimation routines (e.g., METHOD=FIRO), so we must proceed assuming we don't have one.

Examining predictors of differential level-2 random intercept variance across persons:

Level 1:

$$\text{Symptoms}_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Mood}_{ti} - \overline{\text{Mood}}_i\right) + \beta_{2i}\left(\text{Stressor}_{ti}\right) + e_{ti}$$

$$\text{Residual Variance: } \sigma^2_{e_{ti}} = \exp\left[\eta_{0i}\right]$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}\left(\text{Women}_i\right) + \gamma_{02}\left(\text{Age}_i - 80\right) + \gamma_{03}\left(\text{Women}_i\right)\left(\text{Age}_i - 80\right) + \gamma_{04}\left(\overline{\text{Mood}}_i - 2\right)$$
$$+ \gamma_{08}\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{09}\left(\text{Women}_i\right)\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{0,16}\left(\overline{\text{Mood}}_i - 2\right)^2 + U_{0i}$$

$$\text{Within-Person Mood: } \beta_{1i} = \gamma_{10} + \gamma_{14}\left(\overline{\text{Mood}}_i - 2\right)$$

$$\text{Within-Person Stressor: } \beta_{2i} = \gamma_{20} + \gamma_{21}\left(\text{Women}_i\right)$$

$$\text{Random Intercept Variance } \tau^2_{U_{0i}} = \exp\begin{bmatrix} \upsilon_{00} + \upsilon_{01}(\text{Women}_i) + \upsilon_{02}(\text{Age}_i - 80) \\ + \upsilon_{01}(\overline{\text{Mood}}_i - 2) + \upsilon_{08}\left(\overline{\text{Stressor}}_i - 0.40\right) \end{bmatrix}$$

$$\text{Residual Variance: } \eta_{0i} = \varepsilon_{00}$$

```
TITLE1 "Add Level 2 Predictors for Sex, Age, Mood, and Stress";
PROC NLMIXED DATA=example METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS  fint=2.45 fsex=-.33 fage=.07 fsexage=-.09
       fbpmood=-.34 fwpmood=.26 fcstress=1.93 fwpstress=.22
       f2sexstress=-1.33 f1sexstress=-.18 fbpmood2=-1.78 fbpwpmood=.26
       vU0int=-.47 vU0sex=0 vU0age=0 vU0bpmood=0 vU0bpstress=0 varEint=-.50;
* Setting up level-2 equations;
       b0i = fint + fsex*female + fage*age80 + fsexage*female*age80
             + fbpmood*PMmood2 + fcstress*PMstress40
             + f2sexstress*PMstress40*female + fbpmood2*PMmood2*PMmood2 + U0i;
       b1i = fwpmood + fbpwpmood*PMmood2;
       b2i = fwpstress + f1sexstress*female;
* Setting up level-1 equation WITHOUT level-1 residual;
       y = (b0i) + (b1i*WPmood) + (b2i*TVstress0);
* Model for heterogeneous variance at level-2 and level-1;
   vU0 = EXP(vU0int + vU0sex*female + vU0age*age80 + vU0bpmood*PMmood2 + vU0bpstress*PMstress40);
   varE = EXP(varEint);
* Telling it which DV, defining level-1 residual;
* symptom5 is normally distributed with a mean of "y" and a variance of "VarE";
       MODEL symptom5 ~ normal(y,varE);
* Defining random effects: normally distributed with means and variances;
       RANDOM U0i ~ normal([0],[vU0]) SUBJECT=ID;
RUN; TITLE1;
```

Specifications	
Data Set	WORK.EXAMPLE
Dependent Variable	symptom5
Distribution for Dependent Variable	Normal
Random Effects	U0i
Distribution for Random Effects	Normal
Subject Variable	id
Optimization Technique	Newton-Raphson
Integration Method	Adaptive Gaussian Quadrature

Dimensions	
Observations Used	509
Observations Not Used	0
Total Observations	509
Subjects	105
Max Obs Per Subject	5
Parameters	**18**
Quadrature Points	1

NOTE: GCONV convergence criterion satisfied.

Fit Statistics	
-2 Log Likelihood	**1371.0**
AIC (smaller is better)	1407.0
AICC (smaller is better)	1408.4
BIC (smaller is better)	1454.8

The previous model was −2LL(14) = 1379.2. The current model is −2LL(18) = 1371.0, so −2ΔLL(4) = 8.2, p = .085, although only one new effect is significant individually…

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper	Gradient
fint	2.4208	0.3426	104	7.07	<.0001	0.05	1.7415	3.1001	-474E-12
fsex	-0.3364	0.2099	104	-1.60	0.1121	0.05	-0.7527	0.07992	-505E-12
fage	0.07753	0.03087	104	2.51	0.0136	0.05	0.01631	0.1387	4.143E-9
fsexage	-0.08882	0.03427	104	-2.59	0.0109	0.05	-0.1568	-0.02086	3.719E-9
fbpmood	-0.2178	0.7821	104	-0.28	0.7812	0.05	-1.7688	1.3332	4.97E-10
fwpmood	0.2616	0.2212	104	1.18	0.2396	0.05	-0.1770	0.7002	2.09E-12
fcstress	1.8901	0.6084	104	3.11	0.0024	0.05	0.6837	3.0965	1.18E-10
fwpstress	0.2233	0.1710	104	1.31	0.1945	0.05	-0.1157	0.5623	-112E-12
f2sexstress	-1.2330	0.6696	104	-1.84	0.0684	0.05	-2.5608	0.09487	1.08E-10
f1sexstress	-0.1850	0.2052	104	-0.90	0.3692	0.05	-0.5919	0.2218	-107E-12
fbpmood2	-1.5905	0.6760	104	-2.35	0.0205	0.05	-2.9310	-0.2501	-372E-12
fbpwpmood	0.2564	0.3960	104	0.65	0.5188	0.05	-0.5289	1.0416	1.77E-13
vU0int	-1.1082	0.7895	104	-1.40	0.1634	0.05	-2.6739	0.4575	-215E-12
vU0sex	**-0.1760**	**0.3852**	**104**	**-0.46**	**0.6487**	**0.05**	**-0.9398**	**0.5878**	**-211E-12**
vU0age	**0.03867**	**0.03293**	**104**	**1.17**	**0.2430**	**0.05**	**-0.02663**	**0.1040**	**6.8E-10**
vU0bpmood	**-0.7150**	**0.8469**	**104**	**-0.84**	**0.4004**	**0.05**	**-2.3944**	**0.9644**	**1.71E-10**
vU0bpstress	**1.5833**	**0.6031**	**104**	**2.63**	**0.0100**	**0.05**	**0.3874**	**2.7792**	**2.87E-11**
varEint	-0.4968	0.07034	104	-7.06	<.0001	0.05	-0.6363	-0.3573	-133E-12

The effect of $PMstressor40_i$ on the amount of random intercept variance indicates that for every unit higher proportion of stressor days, the log of the random intercept variance is predicted to be higher by 1.58. In other words, there is significantly greater person-to-person variation in mean daily reported physical symptoms among people who report experiencing a greater proportion of stressor days.

Examining predictors of differential level-1 residual variance across persons (retaining all effects of heterogeneity on the level-2 random intercept variance for now):

Level 1 :

$$\text{Symptoms}_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Mood}_{ti} - \overline{\text{Mood}}_i\right) + \beta_{2i}\left(\text{Stressor}_{ti}\right) + e_{ti}$$

Residual Variance: $\sigma^2_{e_{ti}} = \exp\left[\eta_{0i} + \underline{\eta_{1i}\left(\text{Mood}_{ti} - \overline{\text{Mood}}_i\right)} + \underline{\eta_{2i}\left(\text{Stressor}_{ti}\right)}\right]$

Level 2 :

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}\left(\text{Women}_i\right) + \gamma_{02}\left(\text{Age}_i - 80\right) + \gamma_{03}\left(\text{Women}_i\right)\left(\text{Age}_i - 80\right) + \gamma_{04}\left(\overline{\text{Mood}}_i - 2\right)$

$$+ \gamma_{08}\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{09}\left(\text{Women}_i\right)\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{0,16}\left(\overline{\text{Mood}}_i - 2\right)^2 + U_{0i}$$

Within-Person Mood: $\beta_{1i} = \gamma_{10} + \gamma_{14}\left(\overline{\text{Mood}}_i - 2\right)$

Within-Person Stressor: $\beta_{2i} = \gamma_{20} + \gamma_{21}\left(\text{Women}_i\right)$

Random Intercept Variance $\tau^2_{U_{0i}} = \exp\begin{bmatrix}\upsilon_{00} + \upsilon_{01}\left(\text{Women}_i\right) + \upsilon_{02}\left(\text{Age}_i - 80\right) \\ + \upsilon_{04}\left(\overline{\text{Mood}}_i - 2\right) + \upsilon_{08}\left(\overline{\text{Stressor}}_i - 0.40\right)\end{bmatrix}$

Residual Variance:

$$\eta_{0i} = \varepsilon_{00} + \underline{\varepsilon_{01}\left(\text{Women}_i\right)} + \underline{\varepsilon_{02}\left(\text{Age}_i - 80\right)} + \underline{\varepsilon_{04}\left(\overline{\text{Mood}}_i - 2\right)} + \underline{\varepsilon_{08}\left(\overline{\text{Stressor}}_i - 0.40\right)}$$

$$\eta_{1i} = \varepsilon_{10}$$

$$\eta_{2i} = \varepsilon_{20}$$

```
TITLE1 "Level-1 Heterogeneity by L2 for Sex, Age, Mood, and Stress, L1 Mood and Stress";
PROC NLMIXED DATA=example METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS  fint=2.45 fsex=-.33 fage=.07 fsexage=-.09
       fbpmood=-.34 fwpmood=.26 fcstress=1.93 fwpstress=.22
       f2sexstress=-1.33 f1sexstress=-.18 fbpmood2=-1.78 fbpwpmood=.26
       vU0int=-.47 vU0sex=0 vU0age=0 vU0bpmood=0 vU0bpstress=0
       varEint=-.50 varEsex=0 varEage=0 varEbpmood=0 varEbpstress=0 varEwpmood=0 varEwpstress=0;
* Setting up level-2 equations;
       b0i = fint + fsex*female + fage*age80 + fsexage*female*age80
             + fbpmood*PMmood2 + fcstress*PMstress40
             + f2sexstress*PMstress40*female + fbpmood2*PMmood2*PMmood2 + U0i;
       b1i = fwpmood + fbpwpmood*PMmood2;
       b2i = fwpstress + f1sexstress*female;
* Setting up level-1 equation WITHOUT level-1 residual;
       y = (b0i) + (b1i*WPmood) + (b2i*TVstress0);
* Model for heterogeneous variance at level-2 and level-1;
   vU0 = EXP(vU0int + vU0sex*female + vU0age*age80 + vU0bpmood*PMmood2 + vU0bpstress*PMstress40);
   varE = EXP(varEint + varEsex*female + varEage*age80 + varEbpmood*PMmood2
          + varEbpstress*PMstress40 + varEwpmood*WPmood + varEwpstress*TVstress0);
* Telling it which DV, defining level-1 residual;
* symptom5 is normally distributed with a mean of "y" and a variance of "VarE";
       MODEL symptom5 ~ normal(y,varE);
* Defining random effects: normally distributed with means and variances;
       RANDOM U0i ~ normal([0],[vU0]) SUBJECT=ID;
RUN; TITLE1;
```

Dimensions

Observations Used	509
Observations Not Used	0
Total Observations	509
Subjects	105
Max Obs Per Subject	5
Parameters	**24**
Quadrature Points	1

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	**1346.2**
AIC (smaller is better)	1394.2
AICC (smaller is better)	1396.7
BIC (smaller is better)	1457.9

The previous model was −2LL(18) = 1371.0. The current model is −2LL(24) = 1346.2, so −2ΔLL(6) = 24.8, $p < .001$, although only two new effects are significant individually…

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper	Gradient
fint	2.4319	0.3651	104	6.66	<.0001	0.05	1.7079	3.1560	-1.49E-7
fsex	-0.3358	0.2079	104	-1.62	0.1093	0.05	-0.7481	0.07647	-1.43E-7
fage	0.07538	0.03087	104	2.44	0.0163	0.05	0.01417	0.1366	8.042E-7
fsexage	-0.08680	0.03438	104	-2.52	0.0131	0.05	-0.1550	-0.01862	7.225E-7
fbpmood	-0.1622	0.8895	104	-0.18	0.8556	0.05	-1.9262	1.6017	1.228E-7
fwpmood	0.1526	0.3170	104	0.48	0.6311	0.05	-0.4759	0.7812	2.418E-8
fcstress	1.9231	0.5954	104	3.23	0.0017	0.05	0.7425	3.1037	2.812E-8
fwpstress	0.2081	0.1610	104	1.29	0.1989	0.05	-0.1111	0.5273	-2.94E-8
f2sexstress	-1.2523	0.6588	104	-1.90	0.0601	0.05	-2.5586	0.05405	2.74E-8
f1sexstress	-0.1748	0.1950	104	-0.90	0.3721	0.05	-0.5616	0.2120	-2.72E-8
fbpmood2	-1.5408	0.7463	104	-2.06	0.0415	0.05	-3.0208	-0.06075	-1.08E-7
fbpwpmood	0.08266	0.5216	104	0.16	0.8744	0.05	-0.9517	1.1170	-2E-9
vU0int	-1.0869	0.8058	104	-1.35	0.1803	0.05	-2.6848	0.5110	-3.49E-8
vU0sex	-0.1516	0.3839	104	-0.39	0.6938	0.05	-0.9129	0.6098	-3.36E-8
vU0age	0.03924	0.03203	104	1.22	0.2234	0.05	-0.02428	0.1028	2.059E-7
vU0bpmood	-0.7041	0.8717	104	-0.81	0.4211	0.05	-2.4327	1.0245	2.874E-8
vU0bpstress	1.4262	0.5871	104	2.43	0.0169	0.05	0.2619	2.5904	6.579E-9
varEint	-0.08445	0.3079	104	-0.27	0.7844	0.05	-0.6949	0.5260	-6.25E-8
varEsex	**0.07127**	**0.1663**	**104**	**0.43**	**0.6691**	**0.05**	**-0.2585**	**0.4010**	**-4.75E-8**
varEage	**-0.00337**	**0.01311**	**104**	**-0.26**	**0.7979**	**0.05**	**-0.02937**	**0.02264**	**-4.53E-8**
varEbpmood	**0.6972**	**0.3368**	**104**	**2.07**	**0.0409**	**0.05**	**0.02922**	**1.3651**	**2.565E-8**
varEbpstress	**0.6967**	**0.3152**	**104**	**2.21**	**0.0293**	**0.05**	**0.07160**	**1.3219**	**-4.66E-9**
varEwpmood	**0.02650**	**0.3095**	**104**	**0.09**	**0.9319**	**0.05**	**-0.5873**	**0.6403**	**6.281E-9**
varEwpstress	**-0.00774**	**0.2177**	**104**	**-0.04**	**0.9717**	**0.05**	**-0.4395**	**0.4241**	**-2.94E-8**

The effect of $PMmood2_i$ on the amount of residual variance indicates that for every unit higher person mean negative mood, the log of the residual variance is predicted to be higher by 0.6972. Likewise, the effect of $PMstressor40_i$ on the amount of residual variance indicates that for every unit higher proportion of stressor days, the log of the residual variance is predicted to be higher by 0.6967. In other words, there is significantly greater within-person daily fluctuation in reported physical symptoms among people who report greater average levels of negative and who report experiencing a greater proportion of stressor days.

Best model (after trimming any effects that were no longer significant, as shown in Table 8.7):

Level 1 :

$$\text{Symptoms}_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Mood}_{ti} - \overline{\text{Mood}}_i\right) + \beta_{2i}\left(\text{Stressor}_{ti}\right) + e_{ti}$$

Residual Variance: $\sigma^2_{e_{ti}} = \exp\left[\eta_{0i}\right]$

Level 2 :

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}\left(\text{Women}_i\right) + \gamma_{02}\left(\text{Age}_i - 80\right) + \gamma_{03}\left(\text{Women}_i\right)\left(\text{Age}_i - 80\right) + \gamma_{04}\left(\overline{\text{Mood}}_i - 2\right)$

$$+ \gamma_{08}\left(\overline{\text{Stressor}}_i - 0.40\right) + \gamma_{09}\left(\text{Women}_i\right)\left(\overline{\text{Stressor}}_i - 0.40\right) + U_{0i}$$

Within-Person Mood: $\beta_{1i} = \gamma_{10}$

Within-Person Stressor: $\beta_{2i} = \gamma_{20} + \gamma_{21}\left(\text{Women}_i\right)$

Random Intercept Variance $\tau^2_{U_{0i}} = \exp\left[\upsilon_{00} + \upsilon_{08}\left(\overline{\text{Stressor}}_i - 0.40\right)\right]$

Residual Variance:

$$\eta_{0i} = \varepsilon_{00} + \underline{\varepsilon_{04}\left(\overline{\text{Mood}}_i - 2\right)} + \underline{\varepsilon_{08}\left(\overline{\text{Stressor}}_i - 0.40\right)}$$

```
TITLE1 "Final Heterogeneous Variance Model without Quad PM Mood and BP*WP Mood";
PROC NLMIXED DATA=example METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS  fint=2.45 fsex=-.33 fage=.07 fsexage=-.09
       fbpmood=1.66 fwpmood=.26
       fcstress=1.93 fwpstress=.22 f2sexstress=-1.33 f1sexstress=-.18
       vU0int=-.47 vU0bpstress=0
       varEint=-.50 varEbpmood=0 varEbpstress=0;

* Setting up level-2 equations;
      b0ifixed = fint + fsex*female + fage*age80 + fsexage*female*age80
                  + fbpmood*PMmood2 + fcstress*PMstress40
                  + f2sexstress*PMstress40*female;
      b1i = fwpmood;
      b2i = fwpstress + f1sexstress*female;

* Setting up level-1 equation WITHOUT level-1 residual;
      yfixed = (b0ifixed) + (b1i*WPmood) + (b2i*TVstress0);
      y = yfixed + U0i;

* Model for heterogeneous variance at level-2 and level-1;
      vU0 = EXP(vU0int  + vU0bpstress*PMstress40);
      varE = EXP(varEint + varEbpmood*PMmood2 + varEbpstress*PMstress40);

* Telling it which DV, defining level-1 residual;
* symptom5 is normally distributed with a mean of "y" and a variance of "VarE";
      MODEL symptom5 ~ normal(y,varE);
* Defining random effects: normally distributed with means and variances;
      RANDOM U0i ~ normal([0],[vU0]) SUBJECT=ID;
```

```
* Asking for fixed effect predicted values;
      PREDICT yfixed OUT=Predicted;
* Getting all model implied estimates;
      ESTIMATE "Intercept: Men"                fint;
      ESTIMATE "Intercept: Women"              fint + fsex;
      ESTIMATE "Intercept: Sex Diff"           fsex;
      ESTIMATE "Age Slope: Men"                fage;
      ESTIMATE "Age Slope: Women"              fage + fsexage;
      ESTIMATE "Age Slope: Sex Diff"           fsexage;
      ESTIMATE "BP Mood: Linear"               fbpmood;
      ESTIMATE "Contextual Mood: Linear"       fbpmood - fwpmood;
      ESTIMATE "WP Mood"                       fwpmood;
      ESTIMATE "BP Stress: Men"                fcstress + fwpstress;
      ESTIMATE "BP Stress: Women"              fcstress + f2sexstress + fwpstress + f1sexstress;
      ESTIMATE "BP Stress: Sex Diff"           f2sexstress + f1sexstress;
      ESTIMATE "Contextual Stress: Men"        fcstress;
      ESTIMATE "Contextual Stress: Women"      fcstress + f2sexstress;
      ESTIMATE "Contextual Stress: Sex Diff"   f2sexstress;
      ESTIMATE "WP Stress: Men"                fwpstress;
      ESTIMATE "WP Stress: Women"              fwpstress + f1sexstress;
      ESTIMATE "WP Stress: Sex Diff"           f1sexstress;
      ESTIMATE "Contextual Stress per Day: Men"   fcstress/5;
      ESTIMATE "Contextual Stress per Day: Women"(fcstress + f2sexstress)/5;
RUN; PROC CORR DATA=Predicted; VAR symptom5 pred; RUN; TITLE1;
```

Specifications

Data Set	WORK.EXAMPLE
Dependent Variable	symptom5
Distribution for Dependent Variable	Normal
Random Effects	UOi
Distribution for Random Effects	Normal
Subject Variable	id
Optimization Technique	Newton-Raphson
Integration Method	Adaptive Gaussian Quadrature

Dimensions

Observations Used	509
Observations Not Used	0
Total Observations	509
Subjects	105
Max Obs Per Subject	5
Parameters	**15**
Quadrature Points	1

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	**1351.8**
AIC (smaller is better)	1381.8
AICC (smaller is better)	1382.7
BIC (smaller is better)	1421.6

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper	Gradient
fint	2.8017	0.3680	104	7.61	<.0001	0.05	2.0719	3.5314	4.992E-6
fsex	-0.3413	0.2059	104	-1.66	0.1004	0.05	-0.7495	0.06699	3.998E-6
fage	0.06876	0.03007	104	2.29	0.0243	0.05	0.009126	0.1284	-1.21E-6
fsexage	-0.08837	0.03327	104	-2.66	0.0092	0.05	-0.1543	-0.02239	4.658E-7
fbpmood	1.6584	0.3974	104	4.17	<.0001	0.05	0.8703	2.4464	-4.42E-6
fwpmood	0.1024	0.1473	104	0.69	0.4887	0.05	-0.1898	0.3945	-2.69E-9
fcstress	2.0496	0.6011	104	3.41	0.0009	0.05	0.8576	3.2415	-1.48E-6
fwpstress	0.2043	0.1647	104	1.24	0.2176	0.05	-0.1223	0.5309	5.092E-7
f2sexstress	-1.3699	0.6707	104	-2.04	0.0436	0.05	-2.6998	-0.03991	-1.24E-6
f1sexstress	-0.1754	0.1980	104	-0.89	0.3776	0.05	-0.5680	0.2171	3.609E-7
vUOint	-0.5724	0.1712	104	-3.34	0.0011	0.05	-0.9119	-0.2330	-1.16E-6
vUObpstress	1.3881	0.5349	104	2.60	0.0108	0.05	0.3275	2.4488	2.933E-7
varEint	-0.01100	0.2766	104	-0.04	0.9684	0.05	-0.5595	0.5375	-9.49E-7
varEbpmood	0.7228	0.3287	104	2.20	0.0301	0.05	0.07095	1.3746	8.15E-7
varEbpstress	0.6775	0.2250	104	3.01	0.0033	0.05	0.2312	1.1238	2.5E-7

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Intercept: Men	2.8017	0.3680	104	7.61	<.0001	0.05	2.0719	3.5314
Intercept: Women	2.4604	0.3416	104	7.20	<.0001	0.05	1.7829	3.1379
Intercept: Sex Diff	-0.3413	0.2059	104	-1.66	0.1004	0.05	-0.7495	0.06699
Age Slope: Men	0.06876	0.03007	104	2.29	0.0243	0.05	0.009126	0.1284
Age Slope: Women	-0.01961	0.01506	104	-1.30	0.1955	0.05	-0.04947	0.01024
Age Slope: Sex Diff	-0.08837	0.03327	104	-2.66	0.0092	0.05	-0.1543	-0.02239
BP Mood: Linear	1.6584	0.3974	104	4.17	<.0001	0.05	0.8703	2.4464
Contextual Mood: Linear	1.5560	0.4239	104	3.67	0.0004	0.05	0.7153	2.3967
WP Mood	0.1024	0.1473	104	0.69	0.4887	0.05	-0.1898	0.3945
BP Stress: Men	2.2539	0.5774	104	3.90	0.0002	0.05	1.1089	3.3988
BP Stress: Women	0.7086	0.3137	104	2.26	0.0260	0.05	0.08656	1.3306
BP Stress: Sex Diff	-1.5453	0.6402	104	-2.41	0.0175	0.05	-2.8149	-0.2757
Contextual Stress: Men	2.0496	0.6011	104	3.41	0.0009	0.05	0.8576	3.2415
Contextual Stress: Women	0.6797	0.3329	104	2.04	0.0437	0.05	0.01949	1.3399
Contextual Stress: Sex Diff	-1.3699	0.6707	104	-2.04	0.0436	0.05	-2.6998	-0.03991
WP Stress: Men	0.2043	0.1647	104	1.24	0.2176	0.05	-0.1223	0.5309
WP Stress: Women	0.02886	0.1122	104	0.26	0.7975	0.05	-0.1936	0.2513
WP Stress: Sex Diff	-0.1754	0.1980	104	-0.89	0.3776	0.05	-0.5680	0.2171
Contextual Stress per Day: Men	0.4099	0.1202	104	3.41	0.0009	0.05	0.1715	0.6483
Contextual Stress per Day: Women	0.1359	0.06659	104	2.04	0.0437	0.05	0.003897	0.2680

Pearson Correlation Coefficients, N = 509
Prob > |r| under H0: Rho=0

	symptom5	Pred
symptom5	1.00000	0.52894
symptom5: #Physical Symptoms Truncated at 5		<.0001
Pred	**0.52894**	1.00000
Predicted Value	<.0001	

R = .5289, so R^2 for the overall model = .28.

It would be impossible to calculate valid pseudo-R^2 values for each variance component given their heterogeneity across persons and occasions.



Figure 8.3 illustrates the two interactions of stressor days with sex.

Neither sex has a significant WP stress effect (as shown by the slope of the lines), and this non-effect is the same across men and women.

Both men and women show significant effects of reporting more stressor days, even after controlling for current daily stressors. However, these BP and contextual effects of stress are significantly more positive for men (greater difference between lines).